SPIKE

PURPOSE

Specifies if a spike will be drawn at the plot points of each trace on subsequent plots.

DESCRIPTION

A spike is a vertical line from the plot point to the spike base (most typically the x axis).

DATAPLOT can draw a trace as a character or plot symbol at each point, as a connected line, as a spike from the point to a base, as a bar from the plot point to a base, or as any combination of the above. The choice is determined by the BAR, SPIKE, CHARACTER, and LINE commands. The switches for these commands work independently of each other.

Spikes are commonly used in time series plots. They are also useful in showing deviation from a common value such as the mean or median. They are also used to generate dot charts. Dot charts are an alternative to bar charts recommended by Bill Cleveland in the books listed in the REFERENCE section below.

SYNTAX

SPIKE <ON/OFF> <ON/OFF> <ON/OFF> etc.
where ON specifies that the trace is to be drawn with spikes and OFF specifies that it is not. Up to 100 spike types can be specified.

EXAMPLES

SPIKE ON OFF ON OFF
SPIKE ON ALL
SPIKE ALL ON
SPIKE

NOTE 1

Dot charts are typically drawn vertically rather than horizontally. This can done with either the SPIKE DIRECTION or the HORIZONTAL SWITCH command.

NOTE 2

It is common with dot charts to want alphabetic labels. The TIC MARK LABEL FORMAT, TIC MARK LABEL CONTENT, and TIC OFFSET commands can be used for this purpose.

NOTE 3

The SPIKE command with no arguments sets the spike type to blank for all traces. The SPIKE command with the word ALL before or after the specified type assigns that spike type to all traces; thus SPIKE ON ALL or SPIKE ALL ON plots spikes for all traces.

DEFAULT

No spikes are drawn (i.e., all OFF).

SYNONYMS

None

RELATED COMMANDS

PLOT	=	Generates a data or function plot.
SPIKE BASE	=	Sets the base locations for plot spikes.
SPIKE COLOR	=	Sets the colors for plot spikes.
SPIKE DIRECTION	=	Sets the directions for plot spikes.
SPIKE LINE	=	Sets the line types for plot spikes.
SPIKE THICKNESS	=	Sets the line thicknesses for plot spikes.
BAR	=	Sets the on/off switches for plot bars.
CHARACTER	=	Sets the types for plot characters.

REFERENCE

"Elements of Graphing Data," William S. Cleveland, Wadsworth Advanced Books and Software, 1985.

"Visualizing Data," William S. Cleveland, Hobart Press, 1993.

APPLICATIONS

Presentation graphics, time series plots, dot charts

IMPLEMENTATION DATE

Pre-1987

PROGRAM 1

```
SKIP 25
READ BOXJE142.DAT YIELD
.
TITLE TIME SERIES PLOT
TITLE SIZE 5
Y1LABEL YIELD
X1LABEL SEQUENCE NUMBER
XLIMITS 0 70
XTIC OFFSET 2 2
LET N = SIZE YIELD
LET X = DATA 1 N
LET A = MEAN YIELD
LET Y = DATA A A
CHARACTER OFF
SPIKE ON
SPIKE BASE A
LINE BLANK
PLOT YIELD AND
PLOT Y X
```



PROGRAM 2

```
READ STRING S1 S2 S3 S4 S5 S6 S7
ALABAMA ALASKA ARIZONA ARKANSAS CALIFORNIA SP()COLORADO CONNETICUIT
READ STRING S8 S9 S10 S11 S12 S13 S14
DELAWARE WASHINGTONSP()DC FLORIDA GEORGIA HAWAII IDAHO ILLINOIS
READ STRING S15 S16 S17 S18 S19 S20 S21 S22
INDIANA IOWA KANSAS KENTUCKY LOUSIANA MAINE MARYLAND MASSACHUSETS
READ STRING S23 S24 S25 S26 S27 S28 S29
MICHIGAN MINNESOTA MISSISIPPI MISSOURI MONTANA NEBRASKA NEVADA
READ STRING S30 S31 S32 S33
NEWSP()HAMPSHIRE NEWSP()JERSEY NEWSP()MEXICO NEWSP()YORK
READ STRING S34 S35 S36 S37 S38
NORTHSP()CAROLINA NORTHSP()DAKOTA OHIO OKLAHOMA OREGON
READ STRING S39 S40 S41 S42
PENNSYLVANNIA RHODESP()ISLAND SOUTHSP()CAROLINA SOUTHSP()DAKOTA
READ STRING S43 S44 S45 S46 S47 S48 S49 S50 S51
TENNESSE TEXAS UTAH VERMONT VIRGINIA WASHINGTON WESTSP()VIRGINIA
READ STRING S50 S51
WISCONSIN WYOMING
.
SKIP 25
COLUMN LIMITS 5 30
READ MURDER86.DAT RATE
.
TITLE DOT CHART WITH ONE VARIABLE; TITLE SIZE 5
Y1TIC MARK LABEL OFF
YTIC SIZE 0.5
YLIMITS 1 51
YTIC OFFSET 1 1
MAJOR YTIC MARK NUMBER 51
MINOR YTIC MARK NUMBER 0
X1LABEL MURDER RATE (PER 100,000)
XLIMITS 0 30
XTIC OFFSET 0 2
HORIZONTAL SWITCH ON
CHARACTER CIRCLE
CHARACTER FILL ON
CHARACTER HW 0.8 0.7
LINE BLANK
SPIKE ON
SPIKE LINE DOT
LET INDX = SEQUENCE 1 1 51
LET RATE2 = SORTC RATE INDX
PLOT RATE2
HEIGHT 0.7
JUSTIFICATION RIGHT
LET INC = (90-20)/52
LET YPOS = 19.7
LET XPOS = 14
LOOP FOR K = 1 1 51
    LET YPOS = YPOS + INC
    LET J = INDX(K)
    MOVE XPOS YPOS
    TEXT ^S^J
END OF LOOP
```


DOT CHART WITH ONE VARIABLE
WASHINGTON DC
TEXAS
LOUSIANA
NEVADA
FLORIDA
GEORGIA
NEW MEXICO
MICHIGAN
CALIFORNIA
NEW YORK
TENNESSE
ALABAMA
MISSOURI
MARYLAND
ILLINOIS
ALASKA
SOUTH CAROLINA
OKLAHOMA
NORTH CAROLINA
ARKANSAS
ARIZONA
VIRGINIA
COLORADO
KENTUCKY
OREGON
INDIANA
WEST VIRGINIA
PENNSYLVANNIA
OHIO
WYOMING
NEW JERSEY
WASHINGTON
DELAWARE
HAWAII
CONNETICUIT
KANSAS
SOUTH DAKOTA
MASSACHUSETS
RHODE ISLAND
IDAHO
UTAH
WISCONSIN
NEBRASKA
MONTANA
MINNESOTA
MISSISIPPI
NEW HAMPSHIRE
MAINE
VERMONT
IOWA
NORTH DAKOTA
0
5
10
15
20
25
30
MURDER RATE (PER 100,000)

PROGRAM 3

```
SKIP 25
READ BATTADD.DAT WATTS TYPE ID
.
LET TYPE2 = DISTINCT TYPE; LET N1 = SIZE TYPE2
LET ID2 = DISTINCT ID; LET N2 = SIZE ID2
LET X = CODE ID2
LET TEMP = WATTS; RETAIN TEMP SUBSET TYPE = 1; LET INDX = RANK TEMP
.
READ STRING S1 S2 S3
CONTROL AD-X2 MG-NA
LOOP FOR K = 1 1 N2
    LET J = INDX(K)
    LET TX = ID(J)
    LET STRING T^K = ^TX
END OF LOOP
.
HORIZONTAL SWITCH ON
CHARACTER CIRCLE; CHARACTER FILL ON; CHARACTER HW 0.8 0.7
LINE BLANK; SPIKE ON; SPIKE LINE DOT
.
Y1TIC MARK LABEL FORMAT ALPHA
Y1TIC MARK LABEL CONTENT ...
    ^T1 ^T2 ^T3 ^T4 ^T5 ^T6 ^T7 ^T8 ^T9 ^T10 ^T11 ^T12
YLIMITS 1 N2; YTIC OFFSET 1 1
MAJOR YTIC MARK NUMBER N2; MINOR YTIC MARK NUMBER 0
YTIC LABEL SIZE 1.0; YTIC MARK SIZE 0.4
XLIMITS 0 15; XTIC MARK SIZE 1.0; XTIC MARK LABEL SIZE 2.0
.
LEGEND 1 ^S1; LEGEND 1 COORDINATES 10 88
LEGEND 1 JUSTIFICATION RIGHT; LEGEND SIZE 1.5
FRAME CORNER COORDINATES 15 65 85 90
X2TIC MARK ON
X1FRAME OFF
LET TEMP = WATTS; RETAIN TEMP SUBSET TYPE = 1
PLOT TEMP INDX
.
PRE-ERASE OFF
LEGEND 1 ^S2
LEGEND 1 COORDINATES 10 63
FRAME CORNER COORDINATES 15 40 85 65
X2FRAME OFF
LET TEMP = WATTS; RETAIN TEMP SUBSET TYPE = 2
PLOT TEMP INDX
.
LEGEND 1 ^S3
LEGEND 1 COORDINATES 10 38
FRAME CORNER COORDINATES 15 15 85 40
X1FRAME ON
LET TEMP = WATTS; RETAIN TEMP SUBSET TYPE = 3
PLOT TEMP INDX
.
HEIGTH 2.0; JUSTIFICATION CENTER
MOVE 50 97; TEXT BATTERY ACID DATA
MOVE 50 5; TEXT WATT-HOURS
```


BATTERY ACID DATA
CONTROL
22
25
21
19
23
27
26
29
20
30
28
24
AD-X2
22
25
21
19
23
27
26
29
20
30
28
24
MG-NA
22
25
21
19
23
27
26
29
20
30
28
24
0
5
10
15
WATT-HOURS

PROGRAM 4

```
DIMENSION 50 VARIABLES
SKIP 25
READ BATTADD.DAT WATTS TYPE ID
.
LET TYPE2 = DISTINCT TYPE; LET N1 = SIZE TYPE2
LET ID2 = DISTINCT ID; LET N2 = SIZE ID2
LET X = CODE ID2
LET TEMP = WATTS; RETAIN TEMP SUBSET TYPE = 1; LET INDX = SEQUENCE 1 1 N1
.
READ STRING S1 S2 S3
CONTROL AD-X2 MG-NA
LOOP FOR K = 1 1 N2
    LET TX = ID(K)
    LET STRING T^K = ^TX
END OF LOOP
.
HORIZONTAL SWITCH ON
CHARACTER CIRCLE; CHARACTER FILL ON; CHARACTER HW 0.8 0.7
LINE BLANK
SPIKE ON; SPIKE LINE DOT
.
Y1TIC MARK LABEL FORMAT ALPHA
Y1TIC MARK LABEL CONTENT ^S1 ^S2 ^S3
YLIMITS 1 N1; YTIC OFFSET 1 1
MAJOR YTIC MARK NUMBER N1; MINOR YTIC MARK NUMBER 0
YTIC LABEL SIZE 1.0; YTIC MARK SIZE 0.4
XLIMITS 0 15; XTIC MARK SIZE 1.0; XTIC MARK LABEL SIZE 2.0
.
LEGEND 1 JUSTIFICATION RIGHT; LEGEND SIZE 1.5
X2TIC MARK ON
X1FRAME OFF
LET YSTART = 15; LET YSTOP = 90; LET YINC = (YSTOP - YSTART)/N2
LET YBOTTOM = YSTOP
LOOP FOR K = 1 1 N2
    LET YUPPER = YBOTTOM
    LET YBOTTOM = YUPPER - YINC
    FRAME CORNER COORDINATES 15 YBOTTOM 85 YUPPER
    LEGEND 1 ^T^K
    LET YPOS = YUPPER - 2; LEGEND 1 COORDINATES 6 YPOS
    IF K = N2
        X1FRAME ON
    END IF
    LET J = ID2(K)
    LET TEMP = WATTS; RETAIN TEMP SUBSET ID = J
    PLOT TEMP INDX
     .
    PRE-ERASE OFF
    X2FRAME OFF
END OF LOOP
.
HEIGTH 2.0; JUSTIFICATION CENTER
MOVE 50 97; TEXT BATTERY ACID DATA
MOVE 50 5; TEXT WATT-HOURS
```


BATTERY ACID DATA
19
MG-NA
AD-X2
CONTROL
20
MG-NA
AD-X2
CONTROL
21
MG-NA
AD-X2
CONTROL
22
MG-NA
AD-X2
CONTROL
23
MG-NA
AD-X2
CONTROL
24
MG-NA
AD-X2
CONTROL
25
MG-NA
AD-X2
CONTROL
26
MG-NA
AD-X2
CONTROL
27
MG-NA
AD-X2
CONTROL
28
MG-NA
AD-X2
CONTROL
29
MG-NA
AD-X2
CONTROL
30
MG-NA
AD-X2
CONTROL
0
5
10
15
WATT-HOURS

PROGRAM 5

```
DIMENSION 50 VARIABLES
SKIP 25
READ MANDEL.DAT SPECVOL PRESS TEMP TYPE
.
LET PRESS2 = DISTINCT PRESS; LET PRESS2 = SORT PRESS2
LET N1 = SIZE PRESS2
LET TEMP2 = DISTINCT TEMP; LET TEMP2 = SORT TEMP2
LET N2 = SIZE TEMP2
LET TYPE2 = DISTINCT TYPE; LET TYPE2 = SORT TYPE2
LET N3 = SIZE TYPE2
LET INDX = SEQUENCE 1 1 N1
.
FEEDBACK OFF
LOOP FOR K = 1 1 N1
    LET TX = PRESS2(K)
    LET STRING R^K = ^TX
END OF LOOP
LOOP FOR K = 1 1 N2
    LET TX = TEMP2(K)
    LET STRING S^K = ^TX
END OF LOOP
READ STRING T1 T2
PEROXIDE_CURED UNVULCANIZED
.
HORIZONTAL SWITCH ON
CHARACTER CIRCLE; CHARACTER FILL ON; CHARACTER HW 0.8 0.7
LINE BLANK; SPIKE ON; SPIKE LINE DOT
.
YTIC MARK LABEL FORMAT ALPHA
YTIC MARK LABEL CONTENT ^R1 ^R2 ^R3 ^R4 ^R5 ^R6
YLIMITS 1 N1; YTIC OFFSET 1 1
MAJOR YTIC MARK NUMBER N1; MINOR YTIC MARK NUMBER 0
YTIC LABEL SIZE 1.0; YTIC MARK SIZE 0.4; Y1LABEL SIZE 1.2
YTIC MARK LABEL DISPLACMENT 2; YLABEL DISPLACEMENT 7
.
XLIMITS 0 500; XTIC OFFSET 0 50
XTIC MARK SIZE 0.6; XTIC MARK LABEL SIZE 1.2; XTIC LABEL DISPLACEMENT 2
XLABEL SIZE 1.2; XLABEL DISPLACEMENT 5
.
LEGEND 1 JUSTIFICATION RIGHT; LEGEND SIZE 1.5
.
LET YSPACE = 2; LET YSTART = 12; LET YSTOP = 93
LET YINC = (YSTOP - YSTART - (N2-1)*YSPACE)/N2
LET XSPACE = 2; LET XSTART = 10; LET XSTOP = 90
LET XINC = (XSTOP - XSTART - (N3-1)*XSPACE)/N3
LET YBOTTOM = YSTOP + YSPACE
FEEDBACK OFF
.
LOOP FOR K = 1 1 N2
    XTIC MARK OFF; XTIC MARK LABEL OFF
    IF K = 1
        X2TIC MARK ON
    END OF IF
    IF K = N2
        X1TIC MARK ON; X1TIC MARK LABEL ON
    END OF IF
```

```
        LET YUPPER = YBOTTOM - YSPACE; LET YBOTTOM = YUPPER - YINC
        LET XRIGHT = XSTART - XSPACE
         LET YPOS = (YUPPER + YBOTTOM)/2; LET XPOS = XSTOP + 5
        LEGEND 1 COORDINATES XPOS YPOS
        LET J1 = TEMP2(K)
        LOOP FOR KX = 1 1 N3
                LEGEND 1 ^S^K
                YLABEL ; XLABEL ; YTIC LABEL OFF; YTIC OFF
                IF KX = 1
                    Y1LABEL PRESSURE; Y1TIC MARK LABEL ON; Y1TIC MARK ON
                 END OF IF
                IF KX = N3
                    Y2TIC MARK ON
                END OF IF
                IF K = N2
                    XLABEL ^T^KX
                END OF IF
                LET XLEFT = XRIGHT + XSPACE; LET XRIGHT = XLEFT + XINC
                FRAME CORNER COORDINATES XLEFT YBOTTOM XRIGHT YUPPER
                LET J2 = TYPE2(KX)
                LET JUNK = SPECVOL
                RETAIN JUNK SUBSET TEMP = J1 SUBSET TYPE = J2
                PLOT JUNK INDX
                PRE-ERASE OFF
        END OF LOOP
END OF LOOP
.
HEIGTH 2.0; JUSTIFICATION CENTER
MOVE 50 95; TEXT JOHN MANDEL SPECIFIC VOLUME OF RUBBER DATA
MOVE 50 2; TEXT RUBBER TYPE
FONT COMPLEX; ANGLE 90
MOVE 98 50; TEXT TEMPERATURE
```


JOHN MANDEL SPECIFIC VOLUME OF RUBBER DATA
PRESSURE
0
100
200
300
400
500
TEMPERATURE
0
10
20
25
PEROXIDE_CURED
UNVULCANIZED
RUBBER TYPE